

# Anharmonic Oscillator

## ■ Introduction and the simple harmonic oscillator

In this notebook we study some problems in quantum mechanics using matrix methods. We know that we can solve quantum mechanics in any complete set of basis functions. If we choose a particular basis, the Hamiltonian will not, in general, be diagonal, but the task is to diagonalize it to find the eigenvalues (which are the possible results of a measurement of the energy) and the eigenvectors. The reference for this material is Kinzel and Reents, p. 47-51.

In many cases this can not be done exactly and some numerical approximation is needed. A common approach, which is the basis of a lot of quantum chemistry is to take a *finite* basis set and diagonalize it numerically. The ground state of this reduced basis state will not be the exact ground state, but by increasing the size of the basis (up to a point) we can improve the accuracy and see if the energy converges as we increase the basis size. We will apply this approach here for an anharmonic oscillator.

We first discuss the exactly solvable case of the simple harmonic oscillator. The Hamiltonian is given by

$$H_0 = \frac{p^2}{2m} + \frac{1}{2} m \omega^2 x^2$$

where  $p$  is the momentum,  $x$  the position,  $m$  the mass and  $\omega$  the angular frequency of the classical oscillator. This can be written in dimensionless form as

$$\frac{H_0}{\hbar\omega} = \frac{1}{2} \left( \frac{p}{p_0} \right)^2 + \frac{1}{2} \left( \frac{x}{x_0} \right)^2$$

where  $p_0 = \sqrt{\hbar m \omega}$  and  $x_0 = \sqrt{\hbar / (m \omega)}$ , are the basic momentum and length scales. From now on, we will give the energy in units of  $\hbar\omega$ ,  $x$  in units of  $x_0$ , and  $p$  in units of  $p_0$ , so the reduced Hamiltonian is

$$H_0 = \frac{p^2}{2} + \frac{x^2}{2}.$$

In any textbook on quantum mechanics, it is shown that the energy levels are given by

$$E_n = n + \frac{1}{2}, \quad n = 0, 1, 2, \dots$$

and the wavefunctions are given by

$$\psi_n(x) = \left( 2^n n! \sqrt{\pi} \right)^{-1/2} e^{-x^2/2} H_n(x),$$

where  $H_n(x)$  is a Hermite polynomial.

The books also show that it is easier to determine the energy levels using operator methods rather than the Schrödinger equation, and that is the approach that we will take here. In this approach, one introduces so-called raising and lowering operators,  $a^\dagger$  and  $a$ , related to  $x$  and  $p$  by

$$x = \frac{1}{\sqrt{2}} (a^\dagger + a), \quad p = \frac{i}{\sqrt{2}} (a^\dagger - a),$$

which have the commutation relations

$$[a, a^\dagger] = 1.$$

and in terms of which the Hamiltonian is written

$$H_0 = a^\dagger a + \frac{1}{2}.$$

Denoting eigenstates of the Hamiltonian by  $|n\rangle$ , then one finds

$$a^\dagger |n\rangle = \sqrt{n+1} |n+1\rangle, \quad a |n\rangle = \sqrt{n} |n-1\rangle$$

and so

$$a^\dagger a |n\rangle = n |n\rangle,$$

from which it follows that the energy  $E_n$  is equal to  $(n + 1/2)$  as stated above.

The matrix elements of  $x$  in the basis  $|n\rangle$  are given by

$$X_{nm} = \langle n | x | m \rangle = \frac{1}{\sqrt{2}} \sqrt{m+1} \delta_{n,m+1} + \frac{1}{\sqrt{2}} \sqrt{m} \delta_{n,m-1} = \frac{1}{2} \sqrt{n+m+1} \delta_{|n-m|,1}.$$

Hence we can conveniently define the matrix elements for  $x$  as follows:

```
Clear["Global`*"]; Off[General::spell1]
x[n_, m_] := Sqrt[(n+m+1)]/2 /; Abs[n-m] == 1
x[n_, m_] := 0 /; Abs[n-m] != 1
```

and hence generate a matrix for  $x$ , if we keep a basis of "basissize" states:

```
x[basissize_] := Table[x[n, m], {n, 0, basissize-1}, {m, 0, basissize-1}]
```

Lets check that this is correct for basissize = 4

```
x[4] // MatrixForm
```

$$\begin{pmatrix} 0 & \frac{1}{\sqrt{2}} & 0 & 0 \\ \frac{1}{\sqrt{2}} & 0 & 1 & 0 \\ 0 & 1 & 0 & \sqrt{\frac{3}{2}} \\ 0 & 0 & \sqrt{\frac{3}{2}} & 0 \end{pmatrix}$$

It is trivial to generate the Hamiltonian matrix of the simple harmonic oscillator, since it is diagonal, i.e.

```
h0[basissize_] := DiagonalMatrix[Table[n+1/2, {n, 0, basissize-1}]]
```

```
h0[4] // MatrixForm
```

$$\begin{pmatrix} \frac{1}{2} & 0 & 0 & 0 \\ 0 & \frac{3}{2} & 0 & 0 \\ 0 & 0 & \frac{5}{2} & 0 \\ 0 & 0 & 0 & \frac{7}{2} \end{pmatrix}$$

## The anharmonic oscillator

Now we make the problem non-trivial by adding an anharmonic term. We will take it to be proportional to  $x^4$ , i.e.

$$H = H_0 + \lambda x^4$$

It is easy to generate the matrix for H using the matrix obtained above for x and the convenient "dot" notation in *Mathematica* for performing matrix products:

```
h[basissize_, λ_] := h0[basissize] + λ x[basissize] . x[basissize] . x[basissize] . x[basissize]
```

```
h[4, λ] // MatrixForm
```

$$\begin{pmatrix} \frac{1}{2} + \frac{3\lambda}{4} & 0 & \frac{3\lambda}{\sqrt{2}} & 0 \\ 0 & \frac{3}{2} + \frac{15\lambda}{4} & 0 & 3\sqrt{\frac{3}{2}}\lambda \\ \frac{3\lambda}{\sqrt{2}} & 0 & \frac{5}{2} + \frac{27\lambda}{4} & 0 \\ 0 & 3\sqrt{\frac{3}{2}}\lambda & 0 & \frac{7}{2} + \frac{15\lambda}{4} \end{pmatrix}$$


The eigenvalues can also be obtained numerically and then sorted. Here we give a function (with delayed assignment) for doing this:

```
evals[basissize_, λ_] := Sort[Eigenvalues[N[h[basissize, λ]]]]
```

Now we get some numbers. We start with a basis of size 15 and plot the eigenvalues for a range of  $\lambda$ .

```
basissize = 15;
```

```
p1 = Plot[Evaluate[evals[basissize, λ]], {λ, 0, 1}, PlotRange -> {0, 11},  
PlotStyle -> {AbsoluteThickness[2]}, AxesLabel -> {"λ", "E"}]
```




We see that the energy levels and their spacing increase as  $\lambda$  increases. To see how accurate are the energy levels we compare the results for the size-15 basis with those for a size-10 basis.

```
basissize = 10;
```

```
p2 = Plot[Evaluate[evals[basissize, λ]], {λ, 0, 1}, PlotRange -> {0, 10},  
PlotStyle -> {{AbsoluteThickness[2], Hue[0], Dashing[{0.01, 0.02}]}},  
BaseStyle -> {FontSize -> 14, FontWeight -> "Bold", FontColor -> Hue[0]}];
```


```
Show[p1, p2, PlotLabel -> "basissize=15 (solid) and 10 (dashed)"]
```



We see that the lowest levels appear to have converged well, but the higher levels have not. This is not surprising because the highest of the plotted levels are some of the highest possible for the size-10 basis, and so the effects of the truncation of the basis are naturally large in that region.

To be more systematic, we focus on the ground state, which is given by the first element of `evals`. We can plot it as a function of  $\lambda$ :

```
Plot[evals[20,  $\lambda$ ][[1]], { $\lambda$ , 0, 1}]
```




We can also get a numerical value for a particular value of  $\lambda$  (we take 0.2 here)

```
evals[20, 0.2][[1]]
```

```
0.602405
```

and plot it versus the inverse of the size of the basis set. Here are some results:

```
ListPlot[Table[{1/basissize, evals[basissize, 0.2][[1]]}, {basissize, 9, 30}],
Frame → True, Axes → False, PlotStyle → {PointSize[0.02], Hue[0]},
Epilog → Line[{{1/30, 0.6024051508367307`}, {1/8, 0.6024051508367307`}], PlotRange → All,
FrameLabel → {"1/(basis size)", "E0"}, RotateLabel → False, PlotLabel → "λ = 0.2"]
```



We see that for basis size greater than about 16, the energy has converged well. As you might expect, and you can verify yourself, a larger basis set is needed if  $\lambda$  is larger.

The eigenvectors can also be found, and these give the linear coefficients of the Hermite polynomials which make up the coordinate space wavefunction, and hence the wavefunction of the anharmonic oscillator can also be obtained by this method.

In another handout we calculated the lowest three energy levels of this anharmonic oscillator in a completely different way, from the Schrodinger equation. In that approach we solve a differential equation rather than diagonalize a matrix. We found the lowest three energy levels to be

0.602405, 1.95054, 3.5363

We see that the lowest agrees very well with the ground state determined above from matrix methods. To compare all three numbers we now compute the lowest three levels from the matrix method, still with  $\lambda = 0.2$ :

```
{evals[20, 0.2][[1]], evals[20, 0.2][[2]], evals[20, 0.2][[3]]}
{0.602405, 1.95054, 3.5363}
```

All three energies agree with the corresponding result from Schrodinger's equation, to within the digits printed.